

12 Differentiation of Functions of More Than One Variable

12.1 Partial Derivatives with Maple V

Let f be a function of two variables defined around the point (a, b) . Recall that the definition of the partial derivative with respect to x at (a, b) is given by the following limit (provided that the limit exists).

$$\frac{\partial f}{\partial x}(a, b) = \lim_{h \rightarrow 0} \frac{f(a + h, b) - f(a, b)}{h}$$

Similarly, the partial derivative with respect to y at (a, b) is

$$\frac{\partial f}{\partial y}(a, b) = \lim_{k \rightarrow 0} \frac{f(a, b + k) - f(a, b)}{k}.$$

As an example of how Maple V can be used to compute the partial derivatives of the function:

$$f(x, y) = x^4 y^2 - x y + 7$$

at the point (x, y) we first define f and take the appropriate limits.

```
> f := (x, y) -> x^4*y^2 - x*y + 7;
```

$$f := (x, y) \rightarrow x^4 y^2 - x y + 7$$

```
> Limit((f(x+h, y) - f(x, y)) / h, h = 0) := value(");
```

$$\lim_{h \rightarrow 0} \frac{(x+h)^4 y^2 - (x+h) y - x^4 y^2 + x y}{h} = y(4 y x^3 - 1)$$

```
> Limit((f(x, y+k) - f(x, y)) / k, k = 0) := value(");
```

$$\lim_{k \rightarrow 0} \frac{x^4 (y+k)^2 - x (y+k) - x^4 y^2 + x y}{k} = x(2 y x^3 - 1)$$

The preceding Maple V segment tells us that

$$\frac{\partial f}{\partial x}(x, y) = y(4 y x^3 - 1)$$

and

$$\frac{\partial f}{\partial y}(x, y) = x(2 y x^3 - 1)$$

Normally we will simply ask Maple V to compute the partial derivatives directly. We can use the inert form for **diff** if we want to format the output.

```
> Diff(f(x, y), x) := value(");
```

$$\frac{\partial}{\partial x} (x^4 y^2 - x y + 7) = 4 x^3 y^2 - y$$

We need to factor the preceding result in order to see that it agrees with the limit defined above as the partial derivative of f with respect to x .

```
> factor(");
```

$$\frac{\partial}{\partial x} (x^4 y^2 - x y + 7) = y (4 x^3 - 1)$$

Now we will use **diff** to compute the partial with respect to y.

```
> diff(f(x,y),y);
```

$$2 x^4 y - x$$

```
> factor("");
```

$$x (2 y x^3 - 1)$$

It is possible to compute higher order derivatives. For example, the second derivative with respect to x can be obtained as follows:

```
> Diff(f(x,y),x,x): "=value(");
```

$$\frac{\partial^2}{\partial x^2} (x^4 y^2 - x y + 7) = 12 x^2 y^2$$

or

```
> diff(f(x,y),x,x);
```

$$12 x^2 y^2$$

or

```
> diff(f(x,y),x$2);
```

$$12 x^2 y^2$$

In the last step we used the \$ operator which sometimes shortens the notation.

```
> x$5;
```

$$x, x, x, x, x$$

```
> x$3,y$2;
```

$$x, x, x, y, y$$

To compute the fifth order derivative

$$\frac{\partial^5 f}{\partial x^3 \partial y^2}$$

we write:

```
> diff(f(x,y),x$3,y$2);
```

$$48 x$$

or

```
> Diff(f(x,y),x$3,y$2): "= value(");
```

$$\frac{\partial^5}{\partial y^2 \partial x^3} (x^4 y^2 - x y + 7) = 48x$$

It is easy to compute the value of partial derivatives at a particular point, say (3,2), using **subs**.

```
> diff(f(x,y), x$2, y);
```

$$24x^2y$$

```
> subs(x=3,y=2, " ");
```

$$432$$

Exercises 12.1 Find the indicated partial derivatives.

1. $\frac{\partial^3}{\partial x^2 \partial y} (e^{\pi x^2 y})$

2. $\frac{\partial^8}{\partial x^3 \partial y^5} \left(\frac{\cos(x-y)}{x^4} \right)$

12.2 Local Linearity

In the same way that the tangent line at a point can be computed for a function of one variable, one can compute the tangent plane for a function of two variables. The tangent plane for a function $z = f(x, y)$ at the point (a, b) is given by

$$z = f(a, b) + f_x(a, b)(x - a) + f_y(a, b)(y - b).$$

The function given by the right-hand side of the last equation is called the linear approximation of $f(x, y)$ at (a, b) . As an example let

$$g(x, y) = 4 - x^2 - y^2$$

```
> g := (x,y) -> 4-x^2-y^2;
```

$$g := (x, y) \rightarrow 4 - x^2 - y^2$$

Now compute the tangent plane to $z = g(x, y)$ at $(3, 4)$.

```
> gx(3,2) := subs(x=3,y=2,diff(g(x,y),x));
```

$$gx(3,2) := -6$$

```
> gy(3,2) := subs(x=3,y=2,diff(g(x,y),y));
```

$$gy(3,2) := -4$$

```
> LinApprx := (x,y)-> g(3,2)+gx(3,2)*(x-3)+gy(3,2)*(y-2);
```

$$LinApprx := (x, y) \rightarrow g(3, 2) + gx(3, 2)(x - 3) + gy(3, 2)(y - 2)$$

```
> LinApprx(x,y);
```

$$17 - 6x - 4y$$

In Figure 40 you can compare the graph of $g(x, y)$ with its linearization. This figure is constructed by the following Maple V segment.

```
> P1 := plot3d(g(x,y),x=-4..4,y=-4..4,style=patch):";
```

```
> P2 := plot3d(LinApprx(x,y),x=-2..4,y=-2..3):";
```

```
> plots[display3d]({P1,P2},axes=boxed,view=-20..30);
```

In order to get an idea as to how well the linear approximation does its job consider the error

$$|g(x, y) - LinApprx(x, y)|$$

in the approximation where (x, y) is taken from the sequence of points

$$(4, 3), (3.8, 2.8), (3.6, 2.6), (3.4, 2.4), (3.2, 2.2), \text{ and } (3, 2).$$

Using the **seq** we can observe how the error decreases as this sequence of points goes to $(3, 2)$.

```
> seq(evalf(abs(g(4-i/5,3-i/5)-LinApprx(4-i/5,3-i/5))),i=0..5);
```




Figure 40: Paraboloid $g(x, y) = 4 - x^2 - y^2$ and its Tangent Plane at $(3, 2)$.

```
2., 1.280000000, .7200000000, .3200000000, .08000000000, 0
```

The following command illustrates how the **zip** command can be used to obtain the same error terms.

```
> zip((x,y) -> abs(g(x,y)-LinApprx(x,y)),
>      [4,3.8,3.6,3.4,3.2,3],[3,2.8,2.6,2.4,2.2,2]);
[2, 1.28, .72, .32, .08, 0]
```

As one might expect the error gets smaller as (x,y) approaches $(3,2)$.

Recall that another way to find linear approximations to a function of one variable is to use Taylor Polynomials. The next Maple V segment illustrates how the command **mtaylor** can be used to obtain the same linear approximation that was found above.

```
> readlib(mttaylor);
proc() ... end
> mttaylor(g(x,y),[x=3,y=2],2);
```

$$17 - 6x - 4y$$

Thus the multivariable Taylor Polynomial of degree 1 is the same as the linear approximation.

Exercises 12.2 Find the linear approximation of

$$f(x, y) = \sin(x^2 + y^2)$$

near the point $(\sqrt{\frac{\pi}{2}}, \sqrt{\frac{\pi}{2}})$. Make a Maple V plot showing the graph of $f(x, y)$ and its linear approximation.

12.3 Gradients and Directional Derivatives with Maple V

Let f be a function which has partial derivatives around the point (a, b) . Let $U = \langle u_1, u_2 \rangle$ be a unit vector. Then the directional derivative is defined to be the following limit (assuming that it exists):

$$Df(a, b) = \lim_{h \rightarrow 0} \frac{f(a + hu_1, b + hu_2) - f(a, b)}{h}$$

Thus the directional derivative represents the instantaneous rate of change of f as (x, y) moves in the direction $U = \langle u_1, u_2 \rangle$ from the point (a, b) .

As example find the directional derivative of the function

$$g(x, y) = x^2 + y^2 - x \cos(\pi y) - y \sin(\pi x)$$

in the direction $U = \langle 2/\sqrt{5}, 1/\sqrt{5} \rangle$ at the point $(-1, 2)$.

First define the function $g(x, y)$.

```
> g := (x,y) -> x^2+y^2-x*cos(Pi*y)-y*sin(Pi*x);
```

$$g := (x, y) \rightarrow x^2 + y^2 - x \cos(\pi y) - y \sin(\pi x)$$

Now apply the definition

```
> Dg(-1,2) := limit((g(-1+h*2/sqrt(5), 2+h*1/sqrt(5))-g(-1,2))/h, h=0);
```

$$Dg(-1, 2) := -\frac{2}{5}\sqrt{5} + \frac{4}{5}\pi\sqrt{5}$$

That preceding Maple V output gives the rate of change of the values $g(x, y)$ as (x, y) moves in the direction of $\langle 2/\sqrt{5}, 1/\sqrt{5} \rangle$ from the point $(-1, 2)$. There is a more direct way to compute a directional derivative using gradient vectors and vector operations. The function **grad**, gradient is part of the **linalg** package.

```
> with(linalg):
```

```
Warning: new definition for norm
```

```
Warning: new definition for trace
```

We now compute the gradient of g at an arbitrary point:

```
> Gradg := grad(g(x,y), [x,y]);
```

$$Gradg := [2x - \cos(\pi y) - y \cos(\pi x) \pi, 2y + x \sin(\pi y) \pi - \sin(\pi x)]$$

We use **subs** to evaluate the variables at the point $(-1, 2)$.

```
> Gradg := subs(x=-1,y=2,eval(Gradg));
```

$$Gradg := [-2 - \cos(2\pi) - 2\cos(-\pi)\pi, 4 - \sin(2\pi)\pi - \sin(-\pi)]$$

If we want to evaluate the cosine function at its arguments then **eval** must be used again along with **map**.

```
> Gradg := map(s->eval(s), Gradg);
```

$$Gradg := [-3 + 2\pi, 4]$$

We know that the directional derivative of a function $f(x, y)$ at a point (a, b) in the direction $U = \langle u_1, u_2 \rangle$ is also given by the dot product of the gradient of f at (a, b) and the vector U . Hence:

```
> Dg(-1,2) := dotprod(Gradg, [2/sqrt(5), 1/sqrt(5)]);
```

$$Dg(-1, 2) := \frac{2}{5}(-3 + 2\pi)\sqrt{5} + \frac{4}{5}\sqrt{5}$$

```
> Dg(-1,2) := simplify(Dg(-1,2));
```

$$Dg(-1, 2) := -\frac{2}{5}\sqrt{5} + \frac{4}{5}\pi\sqrt{5}$$

If you wish a decimal approximation of this then perform the following.

```
> evalf(Dg(-1,2));
```

4.725424596

Recall that the gradient vector is perpendicular to the level curves. In order to illustrate this, let us draw the gradient vector field of $g(x,y)$ in the plane. This can be done with the command **gradplot** which is part of the plots package.

```
> with(plots):
```

```
> P1 := gradplot(g(x,y), x=-2..0, y=1..3, arrows=SLIM, grid=[30,30]):";
```

Let's plot the level curve $g(x, y) = g(-1, 2)$.

```
P2 := implicitplot(g(x,y)=g(-1,2), x=-2..0, y=1..3):";
```

Now putting both plots together we obtain. See Figure 41

```
display({P1,P2}, scaling=constrained);
```




Figure 41: A Level Curve with Gradient Field




Figure 42: Level Curves with Gradient Field

Note that any place where an arrow from the gradient vector field intersects the level curve the vector is perpendicular to the curve. We will observe this with a few more level curves. See Figure 42

```
> P3:= implicitplot({g(x,y)=g(-0.5,2),g(x,y)=g(-1.5,1.5)}, x=-2..0, y=1..3):";
```

```
> display({P1,P2,P3}, scaling=constrained);
```

Exercises 12.3

1. If $f(x, y, z) = 2xz^2e^{z^2} - x^2 \sin y^2$, find the gradient of f and its magnitude at $(1, 1, \sqrt{\pi/2})$.
2. If $f(x, y, z) = 2xz^2e^{z^2} - x^2 \sin y^2$, find the directional derivative of f at $(1, 1, \sqrt{\pi/2})$ in the direction parallel to the line joining $(1, 2, 3)$ to $(-3, 3, 6)$.

12.4 Quadratic Approximations

Just as one can use the Maple V procedure **mtaylor** to linearly approximate functions of more than one variable, one can also find higher order approximations of these same functions. Using **mtaylor** it is easy to obtain Taylor Polynomials of degree 2. The following Maple V segment shows how one can find the quadratic approximation for the function

$$f(x, y) = \sin x \cos y$$

near the point $(0, 0)$. First we make the procedure **mtaylor** available with a call to **readlib**.

```
> readlib(mttaylor);
proc() ... end
Now define the function f.
> f := (x,y)-> sin(x)*sin(y);
```

$$f := (x, y) \mapsto \sin(x) \sin(y)$$

Now invoke the procedure and assign the approximation to the variable $f2$.

```
> f2 := mttaylor(f(x,y), [x=0,y=0], 3);
```

$$f2 := xy$$

Next convert the expression $f2$ to a function with the procedure **unapply**.

```
> f2 := unapply(f2,x,y);
```

$$f2 := (x, y) \mapsto xy$$

Some graphical comparisons of the function f with its quadratic approximation $f2$ will now be made. Plots of f and $f2$ are created by the following two Maple V commands and shown in Figures 43 and 44 respectively.

```
> P1 := plot3d(f(x,y), x=-Pi..Pi, y=-Pi..Pi, orientation=[45,60],
> style=PATCHCONTOUR);
> P2 := plot3d(f2(x,y), x=-Pi..Pi, y=-Pi..Pi, orientation=[45,60],
> style=PATCHCONTOUR );
```

In order to illustrate that the quadratic approximation is a pretty good estimate for $f(x, y)$




Figure 43: Plot of $\sin x \cos y$




Figure 44: Plot of Quadratic Approximation of $\sin x \cos y$

near the origin the Maple V **plots[display]** is used to display both functions simultaneously in Figure 45 over the small square $[-0.5, 0.5] \times [-0.5, 0.5]$.

```
> plots[display]({P1,P2},view=[-.5..0.5,-.5..0.5,-1..1]);
```

The same functions are plotted over the larger square $[-\pi, \pi] \times [-\pi, \pi]$. See Figure 46.

```
> plots[display]({P1,P2});
```




Figure 45: Plot of $\sin x \cos y$ and its Quadratic Approximation in a small neighborhood of $(0, 0)$




Figure 46: Plot of $\sin x \cos y$ and its Quadratic Approximation in a larger neighborhood of $(0, 0)$

Plotting contours of these two functions illustrate how the approximation is pretty good near the origin, but that global features of the function of the function f are lost further away from the origin. See Figure 47 for a contour plot of the function f . Compare this plot with Figure 48 which shows a contour plot of the quadratic approximation f_2 .

```
> plot3d(f(x,y),x=-Pi..Pi,y=-Pi..Pi,orientation=[270,0],
> style=contour);
> plot3d(f2(x,y),x=-Pi..Pi,y=-Pi..Pi,orientation=[270,0],
> style=contour);
```




Figure 47: Contour Plot of $\sin x \cos y$ in a neighborhood of $(0, 0)$




Figure 48: Contour Plot of Quadratic Approximation in a neighborhood of $(0, 0)$

Exercises 12.4 Use **mtaylor** to find linear and quadratic approximations near the origin for each of the following functions. Make graphical comparisons of each.

1. $f(x, y) = e^{3x^2} \cos y$

2. $f(x, y) = \sin(x - y)$

3. $f(x, y) = \frac{1}{1-x+y}$

12.5 Partial Differential Equations

Because of its ability in helping to find partial derivatives, Maple V can be used to verify when a function is a solution of a partial differential. In addition, using graphics one can study qualitative features of the solutions. As an example, consider the heat equation

$$\frac{\partial}{\partial t} u = \frac{\partial^2}{\partial x^2} u$$

Using Maple V it easy to show that the function

$$u(x, t) = -e^{(-4t)} \sin(2x)$$

is a solution.

```
> u := (x,t) -> -exp(-4*t)*sin(2*x);
```

$$u := (x, t) \rightarrow -e^{(-4t)} \sin(2x)$$

Now differentiating we see that u satisfies the heat equation.

```
> LeftHandSide := diff(u(x,t),t);
```

$$LeftHandSide := 4e^{(-4t)} \sin(2x)$$

```
> RightHandSide := diff(u(x,t),x,x);
```

$$RightHandSide := 4e^{(-4t)} \sin(2x)$$

Suppose, for example, that $u(x, t)$ represents the heat in a bar which is π units long. Initially the heat at point x along the bar is $u(x, 0)$.

```
> u(x,0);
```

$$-\sin(2x)$$

We can plot this to obtain a graph of the temperature distribution along the bar at times $t = 0$ and $t = 1/8$, respectively. See Figures 49 and 50.

```
> P.1 := plot(u(x,0),x=0..Pi):";
```

```
> P.2 := plot(u(x,1/8),x=0..Pi):";
```

Its rather difficult to see how one plot differs from the other, but if we place both plots on the same graph we get an idea about how the temperature changes as t varies from 0 to $1/8$. See Figure 51

```
> with(plots):
```

```
> display([P.1,P.2]);
```

In order to view changes as t changes from $2/8$ to $7/8$ in increments of $1/8$ we perform the following **do** loop.





```
> for i from 3 to 8 do
```

```
> P.i := plot(u(x,(i-1)/8),x=0..Pi):
```

```
> od:
```

Now we can get a multiple plot that shows how the heat varies as t changes from 0 to $7/8$ in increments of $1/8$. See Figure 52

```
> display([seq(P.i,i=1..8)]);
```

Figure 49: Graph of $u(x, 0)$ Figure 50: Graph of $u(x, 1/8)$ Figure 51: Graph of $u(x, 0)$ and $u(x, 1/8)$ Figure 52: Multiplot of $u(x, i/8)$, $i = 0 \dots 7$

This same information can be animated by using the option **insequence = true**. Thus we obtain an animated plot showing how the temperature in the bar varies as t varies from 0 to $7/8$ in increments of $1/8$.

```
> display([seq(P.i,i=1..8)],insequence=true);
```

Exercises 12.5

1. A twice continuously differentiable function $u(x, y)$ which satisfies the Laplace's Differential Equation

$$\frac{\partial^2 u}{\partial x^2} + \frac{\partial^2 u}{\partial y^2} = 0$$

is said to be a harmonic function. Determine which of the following functions are harmonic.

- (a) $u(x, y) = e^{\pi x} \sin(\pi y)$
- (b) $u(x, y) = x^3 + y^3$
- (c) $u(x, y) = \ln(x^2 + y^2)$
- (d) $u(x, y) = \arctan(y/x)$

2. Verify that

$$y(x, t) = \frac{3}{40} \cos(2t) \sin(x) - \frac{1}{40} \cos(6t) \sin(3x)$$

satisfies the wave equation

$$\frac{\partial^2 y}{\partial t^2} = 4 \frac{\partial^2 y}{\partial x^2}, \quad (0 < x < \pi, \ t > 0,)$$

and the boundary conditions

$$\begin{aligned} y(0, t) &= y(\pi, t) = 0, \\ y(x, 0) &= \frac{3}{40} \sin(x) - \frac{1}{40} \sin(3x). \end{aligned}$$

12.6 Local and Global Extrema

In this section the problem of finding extreme values of functions of more than one variable will be studied. Consider the function f defined as:

$$f(x, y) = xye^{(-1/2x^2 - 1/2y^2)}$$

The function can be defined into a Maple V session by the following command.

```
> f := (x, y) -> x*y*exp(-(x^2+y^2)/2);
```

$$f := (x, y) \rightarrow xye^{(-1/2x^2 - 1/2y^2)}$$

To get an idea about extreme values consider what a graph of this function shown in Figure 53 looks like.




Figure 53: Graph of $xye^{-(x^2+y^2)/2}$




Figure 54: Contourplot of $xye^{-(x^2+y^2)/2}$

```
> plot3d(f(x,y), x=-3..3, y=-3..3, style =patch, orientation=[70, 65],
> axes=FRAMED);
```

This plot suggests that there appears to be an extreme value in each quadrant. There appears to be minima in the second and fourth quadrants and maxima in the first and third quadrants.

The contour plot shown in Figure 54 is useful in finding the approximate location of critical points.

```
> with(plots):
> contourplot(f(x,y), x=-3..3, y=-3..3);
```

In order to find the exact location of the critical points for the function f analytically we need to determine the first derivatives.

```
> fx := diff(f(x,y), x);
```

$$fx := ye^{(-1/2x^2 - 1/2y^2)} - x^2 ye^{(-1/2x^2 - 1/2y^2)}$$

The expression f_x can be simplified.

```
> fx := factor(fx);
```

$$fx := -ye^{(-1/2x^2 - 1/2y^2)}(x-1)(x+1)$$

```
> fy := factor(diff(f(x,y), y));
```

$$f_y := -xe^{(-1/2x^2-1/2y^2)}(y-1)(y+1)$$

The critical points are the points for which f_x and f_y vanish simultaneously.

```
> SOL := solve({fx=0,fy=0},{x,y});
```

$$SOL := \{x=0, y=0\}, \{x=1, y=1\}, \{x=1, y=-1\}, \{x=-1, y=1\}, \\ \{x=-1, y=-1\}$$

Thus there are five critical points:

$$(0, 0), (1, 1), (1, -1), (-1, 1), \text{ and } (-1, -1).$$

Inspection of the plots made above suggest that the points $(1, 1)$ and $(-1, -1)$ are maxima, and points $(-1, 1)$, $(1, -1)$ are minima. Finally, the plots also suggest that $(0, 0)$ is a saddle. In order to get a better focus on this plot the surface in a neighborhood of each critical point.

First check out a neighborhood of $(0, 0)$. See Figure 55.




Figure 55: Graph of $xye^{-(x^2+y^2)/2}$ near $(0, 0)$




Figure 56: Graph of $xye^{-(x^2+y^2)/2}$ near $(1, 1)$

```
> plot3d(f(x,y),x=-0.1..0.1,y=-0.1..0.1,orientation = [70,65],
> style=patchcontour);
```

Indeed this Figure 55 provides further evidence that the point $(0, 0)$ is a saddle. Now check the neighborhood of the point $(1, 1)$ which the previous plots suggest to be a maximum. See Figure 56.

```
> plot3d(f(x,y),x=0.9..1.1,y=0.9..1.1,orientation = [70,65]);
```

This point looks like a maximum. Finally, a plot around $(-1,1)$ should also look like a minimum.

```
> plot3d(f(x,y),x=-1.1..-0.9,y=0.9..1.1,orientation = [70,65],
> style=patchcontour);
```

In the same way, one can obtain geometric evidence by plotting around the other two critical points, but instead we will use analytic means to characterize the critical points.

Recall that if (a, b) is a critical point of f and

$$\Delta = f_{xx}(a, b)f_{yy}(a, b) - f_{xy}(a, b)^2,$$

then

- (a) If $\Delta > 0$ and $f_{xx}(a, b) > 0$, (a, b) is a minimum.
- (b) If $\Delta > 0$ and $f_{xx}(a, b) < 0$, (a, b) is a maximum.
- (c) If $\Delta < 0$, then (a, b) is a saddle point.
- (d) If $\Delta = 0$, anything can happen.

```
> fxx := factor(diff(f(x,y),x,x));
```

$$f_{xx} := x y e^{(-1/2 x^2 - 1/2 y^2)} (-3 + x^2)$$

```
> fxy := factor(diff(f(x,y),x,y));
```

$$f_{xy} := e^{(-1/2 x^2 - 1/2 y^2)} (x - 1)(x + 1)(y - 1)(y + 1)$$

```
> fyy := factor(diff(f(x,y),y,y));
```

$$f_{yy} := x y e^{(-1/2 x^2 - 1/2 y^2)} (-3 + y^2)$$

```
> Delta := factor(fxx*fyy-fxy^2);
```

$$\Delta := -(e^{(-1/2 x^2 - 1/2 y^2)})^2 (-5 x^2 y^2 + x^2 y^4 + x^4 y^2 + 1 - 2 x^2 - 2 y^2 + x^4 + y^4)$$

We now apply the test using the critical point $(0, 0)$:

```
> eval(subs(x=0,y=0,Delta));
```

$$-1$$

Thus $\Delta < 0$ at the point $(0, 0)$ and this point is a saddle point.

Now check the values of f_{xx} and Δ at the point $(1, 1)$.

```
> subs(x=1,y=1,[fxx,Delta]);
```

$$[-2e^{(-1)}, 4(e^{(-1)})^2]$$

Since the number e is positive the last Maple V output means that $f_{xx} < 0$, and $\Delta > 0$ at $(1, 1)$, which proves that $(1, 1)$ is the point where the maximum occurs. We can use **evalf** to approximate the values.

```
> evalf(");
```

$$[-.7357588824, .5413411332]$$

Next a check of the critical point $(-1, 1)$ is made.

```
> subs(x=-1,y=1,[fxx,Delta]);
```

$$[2e^{(-1)}, 4(e^{(-1)})^2]$$

Thus $(-1, 1)$ is a minimum.

Finally, for $(-1, -1)$ and $(1, -1)$.

```
> subs(x=-1,y=-1,[fxx,Delta]);subs(x=1,y=-1,[fxx,Delta]);
```


$$[-2e^{(-1)}, 4(e^{(-1)})^2]$$

$$[2e^{(-1)}, 4(e^{(-1)})^2]$$

You may conclude that $(-1, -1)$ is a maximum and $(1, -1)$ is a minimum.

The following Maple V segment shows how to obtain the value of $f(x, y)$ at the four extreme values using **zip**:

```
> zip(f, [1, -1, -1, 1], [1, 1, -1, -1]);
```

$$[e^{(-1)}, -e^{(-1)}, e^{(-1)}, -e^{(-1)}]$$

The approximate extreme values to three decimal places are as follows:

```
> evalf(", 3);
```

$$[.368, -.368, .368, -.368]$$

The following table summarizes the results for this problem.

(a,b)	$f_{xx}(a, b)$	Δ	Conclusion	$f(x,y)$
(0,0)	0	Negative	Saddle Point	0
(1,1)	Negative	Positive	Maximum	.368
(-1,1)	Positive	Positive	Minimum	-.368
(-1,-1)	Negative	Positive	Maximum	.368
(1,-1)	Positive	Positive	Minimum	-.368

Exercises 12.6 Find and classify all critical points for the following functions of two variables. If an extreme value is absolute then state why it is.

1. $f(x, y) = xy - x^2 - y^2 - 2x - 2y + 4$
2. $f(x, y) = x^3/3 + 9y^3 - 4xy$
3. $f(x, y) = e^{-2x} \cos y$
4. $f(x, y) = xy + 2x - \ln x^2 y, \quad x > 0, y > 0$

12.7 Constrained Optimization

We illustrate how Maple V can be used to solve optimization problems that have constraints with the following example.

Example: Find the dimensions of the box of largest volume which can be fitted inside the ellipsoid

$$\frac{x^2}{a^2} + \frac{y^2}{b^2} + \frac{z^2}{c^2} = 1$$

assuming that each edge of the box is parallel to a co-ordinate axis.

Solution: Since each of the eight vertices of the box must lie on the ellipsoid, let the vertex in the first octant have coordinates (x, y, z) . The dimensions of the box are $2x$, $2y$, $2z$, and its volume is $V = 8xyz$. The problem is to find the values of (x, y, z) , that lie on the ellipsoid and that make V as large as possible.

Thus we wish to optimize $V = 8xyz$ subject to the constraint:

$$g(x, y, z) = 1$$

where

$$g(x, y, z) = \frac{x^2}{a^2} + \frac{y^2}{b^2} + \frac{z^2}{c^2}.$$

Thus we must solve the equations given by

$$\text{grad}(V) = \lambda \text{grad}(g),$$

for x, y, z , and λ , subject to the above constraint. First define V and g using Maple V.

```
> V := 8*x*y*z;
```

$$V := 8xyz$$

```
> g := x^2/a^2+y^2/b^2+z^2/c^2;
```

$$g := \frac{x^2}{a^2} + \frac{y^2}{b^2} + \frac{z^2}{c^2}$$

```
> with(linalg):
```

Warning: new definition for norm

Warning: new definition for trace

The Maple V procedure **grad** from the linear algebra package can be used to compute the necessary gradients.

```
> GradV := grad(V,[x,y,z]);
```

$$\text{Grad}V := [8yz \ 8xz \ 8xy]$$

```
> Gradg := grad(g,[x,y,z]);
```

$$\text{Grad}g := \left[2\frac{x}{a^2} \ 2\frac{y}{b^2} \ 2\frac{z}{c^2} \right]$$

Now one can solve the relevant equations for (x, y, z) and λ .

```
> SOL := solve({g=1, seq(GradV[i]=lambda*Gradg[i], i=1..3)}, {x,y,z,lambda});
```

$$\begin{aligned}
SOL := & \{x = 0, z = 0, \lambda = 0, y = b\}, \{x = 0, z = 0, \lambda = 0, y = -b\}, \\
& \left\{z = \%1, y = \frac{\%1 b}{c}, x = -\frac{\%1 a}{c}, \lambda = -4 \%1 a b\right\}, \\
& \left\{z = \%1, \lambda = -4 \%1 a b, y = -\frac{\%1 b}{c}, x = \frac{\%1 a}{c}\right\}, \\
& \left\{z = \%1, y = \frac{\%1 b}{c}, x = \frac{\%1 a}{c}, \lambda = 4 \%1 a b\right\}, \\
& \left\{z = \%1, x = -\frac{\%1 a}{c}, y = -\frac{\%1 b}{c}, \lambda = 4 \%1 a b\right\}, \\
& \{z = 0, y = 0, \lambda = 0, x = a\}, \{z = 0, y = 0, \lambda = 0, x = -a\}, \\
& \{x = 0, y = 0, \lambda = 0, z = c\}, \{x = 0, y = 0, \lambda = 0, z = -c\} \\
& \%1 := \text{RootOf}(3_Z^2 - c^2)
\end{aligned}$$

Warning: Maple V may return its output in this case in a variety of different but equivalent ways. For instance, if you have typed in the very same commands as was done above, you may get different looking answers. In this particular segment, Maple has returned ten sets of answers. We must now choose those that are correct. Since we are seeking values of x , y , and z that are positive we can exclude all answers that cause these values to be otherwise. First note that the symbol $\%1$ is defined in terms of **RootOf**, a quadratic expression.

```
> allvalues(%1);
```

$$\frac{1}{3}\sqrt{3}c, -\frac{1}{3}\sqrt{3}c$$

If we select the positive roots in each case and inspect the various ten sets of points in SOL above we see that only one answer is eligible to be correct:

$$z = \%1, y = \%1b/c, x = \%1a/c, \lambda = 4\%1ab.$$

Remember that you may get a slightly different set of symbols here, but the results should still amount to the same. Thus the desired values of x_{opt} , y_{opt} , and z_{opt} are

```
> xopt:= allvalues(%1)[1]*a/c; yopt := allvalues(%1)[1]*b/c;
```

$$x_{opt} := \frac{1}{3}\sqrt{3}a$$

$$y_{opt} := \frac{1}{3}\sqrt{3}b$$

```
> zopt :=allvalues(%1)[1];
```

$$z_{opt} := \frac{1}{3}\sqrt{3}c$$

With these values assigned we conclude that the maximum volume is

```
> subs(x=xopt,y=yopt,z=zopt,V);
```

$$\frac{8}{9}ab\sqrt{3}c$$

And with these values substituted into g we see that indeed (x, y, z) satisfy the constraint.

```
> subs(x=xopt,y=yopt,z=zopt,g);
```

1

Exercises 12.7

1. Find the maximum of the function $z = -4x^3y^2$ where (x, y) must be a point on the unit circle centered at the origin.
2. Given $a^x b^y c^z = A$, find the maximum value of

$$(x + 1)(y + 1)(z + 1).$$

3. Find the maximum of

$$x^m y^n z^p$$

subject to the constraint

$$x + y + z = a.$$

4. The Cobb-Douglas production model for a manufacturing process depending three inputs x , y , and z with unit costs a , b , and c , respectively is given by

$$P = kx^\alpha y^\beta z^\gamma, \quad \alpha > 0, \beta > 0, \gamma > 0, \alpha + \beta + \gamma = 1$$

subject to the constraint

$$ax + by + cz = d.$$

Determine x , y , and z to maximize P .